

Inducing Probabilistic Programs by Bayesian Program Merging

Abstract

This report outlines an approach for representing generative models with probabilistic programs and learning these programs from data. The main advantage of using probabilistic programs as generative models is the ability to capture interesting patterns in a concise manner that is amenable to inference. Formulating model learning in terms of program induction also leads us to explore the idea of understanding pattern as repeated computation within a generative model. This idea provides a unifying theme for the program transformations used in searching the space of generative models, which is framed in a Bayesian model merging-like way. There are two main types of program transformations we consider: the first is based on merging common subexpressions within a program and the second focuses on reducing the number of parameters for functions in the program. We demonstrate this approach on a simple domain of colored trees.

1 Introduction

What do you see when you look at figure 1? What kinds of patterns are there in the image?



Figure 1: Some plants.

You might describe the image as a series of plants where each plant has a large base and a number of branches of variable length with each branch ending in a flower that is either light or dark red. Each part of this description corresponds to a different pattern detected in the image. The large base for each individual plant, the variable length branches, branches end in red flowers, etc. Recognizing patterns like these is an important aspect of intelligence and it would be useful if we could automate this process. One possible way to approach this problem is in terms of learning generative models represented as probabilistic programs. In this document, we build on the notion of representing patterns or concepts as probabilistic programs [TODO cite stuhlmueeller et al] and begin to explore a family of algorithms for learning such programs.

One of the major difficulties with identifying regularities in data is the many notions of what it means to be a pattern and how this can vary with the types of data being analyzed. We propose approaching this problem by performing a “reduction” on pattern recognition problems, first transforming data (whatever its type) into some canonical form (an expression in a programming language) and then using the notion of pattern as repeated computation in the program. The main components of our approach are as follows: representation of data in terms of algebraic data types, representation of patterns in data as probabilistic programs, guiding search through program space using a Bayesian posterior probability, and creating search moves based on detecting repeated computation. The general algorithm can be summarized as 1) turn the data into a large program 2) identify repeated computation in the large program and extract these patterns by merging different parts of the program 3) determine which merge moves to apply based on the posterior probability distribution of the program given the data. The probabilistic programs learned in such a manner can be understood as generative models and reasoning about such

models can be formulated in terms of probabilistic inference. We illustrate these ideas on list structured data.

Generative models play a prominent role in modern machine learning and understanding different classes of models e.g. hidden Markov models, probabilistic context-free grammars, etc. have led to a wide variety of applications. There is a trade-off between the variety of patterns a model class is able to capture and the feasibility of learning models in that class [TODO add gulwani citation? or AIMA]. Much of machine learning has focused on studying classes of models with limited expressiveness in order to develop tractable algorithms for modeling large data sets. Our investigation takes a different approach and explores how learning might proceed in a very expressive class of models with a focus on identifying patterns from very small amounts of data.

We examine a class of generative models that are represented as programs in a probabilistic programming language. These programs can have parameterized functions and recursion, which allow for natural representation of “long-range” dependencies and recursive patterns. We will frame searching this space of models in terms of Bayesian model merging [TODO add stolcke et al citation] and demonstrate the class’ ability to capture interesting patterns in the list data domain.

Data is generated from such a model by evaluating the program. Another way to view it is each program represents some probability distribution and each evaluation of the program results in a sample from the distribution. We implement these programs in a subset of the probabilistic programming language Church [add citation goodman et al].

1.1 Bayesian Model Merging

Bayesian model merging is a framework for searching through a space of generative models in order to find one that accurately generates the desired data. The main idea is search program space through a series of merge transformations and to use the posterior $P(M|D)$ as a criteria for selecting transformations, here M is the model and D is the training data. The method works by creating an initial model that has a uniform distribution over the training set by explicitly incorporating each training point into the model (details are model class specific), this process is called data incorporation. While the initial model has a good likelihood $P(D|M)$, it will never generate points outside the training set, in other words it severely overfits the initial data. Alternative model hypotheses are explored by transforming the initial model with merge operations. Merge operations collapse the structure of a model (again specific details are dependent on the model class) and the result may have better generalization properties. This technique has been successfully applied to learning artificial probabilistic grammars where the grammars were represented as hidden Markov models, n-grams, and probabilistic context-free grammars. Our work also uses the posterior distribution as a search criteria and merging model structure to achieve generalization. Our work differs in that we use a significantly richer model class of probabilistic programs. This allows for more complex patterns to be represented as well allowing for more types of transformations that lead to interesting generalization of the model, which can be thought of as lossy compression. The use of programs as generative models also lends itself to a deeper understanding of the relationship between pattern, computation, and intelligence.

2 Overview

The main parts of Bayesian program merging are as follows:

Data Data is represented in terms of some algebraic data type, which gives us a way to form initial programs using the type constructors.

Language Programs are written in a formal language that consists of type constructors and additional operators.

Search Search is based on identifying programs with a high posterior probability and then applying transformations to those programs to continue exploration.

Merge Operations Transformations on programs merge program structure, which can increase generalization of the model as well as the prior probability.

3 Data Representation and Algebraic Data Types

We assume the data we deal with can be modeled with an algebraic data type whose constructors are known to us. This assumption gives us a starting point for program induction since any data can be directly translated into a program which is a derivation (sequence of constructor operations) of the data from the type specification.

3.0.1 Lists Example

Looking back to figure 1 we can model the plants in terms of lists or s-expressions. Each plant consists of nodes, where each node has a size and color along with a list of child nodes. This follows the classic recursive definition of a list below.

$\langle Tree \rangle := \text{'(node' } \langle data \rangle \langle Tree \rangle \langle Tree \rangle \dots \text{')'}$

$\langle data \rangle := \text{'(data' } \langle color \rangle \langle size \rangle \text{')'}$

$\langle color \rangle := \text{'(color' } \langle number \rangle \text{')'}$

$\langle size \rangle := \text{'(size' } \langle number \rangle \text{')'}$

Figure 3.0.1 contains both the graphical and list representation of a plant as well as its specification in terms of the node data type.

```
(node (data (color 50) (size 1))
      (node (data (color 30) (size 0.3))
            (node (data (color 225) (size 0.3)))
            (node (data (color 225) (size 0.3)))
            (node (data (color 225) (size 0.3)))))
```



Figure 2: A plant. The reds are actually different!

3.0.2 Data Incorporation


The first step of Bayesian program merging is data incorporation. Data incorporation is the creation of an initial model by going through each example in the training set and creating an expression in terms of the algebraic data type constructors that evaluates to it. These programs are combined into a single expression that draws uniformly over them. In the case of the plant example we can define it as follows

```
(define (incorporate-data data)
  (let* ([plant-expressions (map plant->expression data)])
    '(lambda () ((uniform-draw (list ,@(map thunkify plant-expressions)))))))


(define (tree->expression tree)
  (if (null? tree)
      '()
      '(node ,(node-data->expression (first tree)) ,(map tree->expression (rest tree)))))

(define (node-data->expression lst)
```

```
(define (thunkify sexpr) '(lambda () ,sexpr))
```

$$\begin{aligned} &(((70) (1)) \\ &\quad ((37) (0.3)) ((213) (0.3))) \\ &\quad ((207) (0.3)) ((211) (0.3)))) \\ &((43) (1)) \\ &\quad ((47) (0.1)) \\ &\quad ((33) (0.3)) ((220) (0.3))) \\ &\quad ((224) (0.3)) ((207) (0.3)))))) \end{aligned}$$


```
(lambda ()
  ((uniform-draw
    (list
      (lambda ()
        (node (data (color 70) (size 1))
          (node (data (color 37) (size 0.3))
            (node (data (color 213) (size 0.3)))
            (node (data (color 207) (size 0.3)))
            (node (data (color 211) (size 0.3)))))))
      (lambda ()
        (node (data (color 43) (size 1))
          (node (data (color 47) (size 0.1))
            (node (data (color 33) (size 0.3))
              (node (data (color 220) (size 0.3)))
              (node (data (color 224) (size 0.3)))
              (node (data (color 207) (size 0.3)))))))))))
```



3.1 Patterns and Probabilistic Programs

4

```

(define (plant)
  (let* ([num-branches (uniform-draw '(1 2 3 4))]
        [branches (repeat num-branches branch)])
    (body branches)))

(define (body branches)
  (pair (data (color 50) (size 1)) branches))

(define (branch)
  (if (flip)
    (flower (if (flip) 150 255))
    (node (data (color 0) (size .1)) (branch))))

(define (flower shade)
  (node (data (color 0) (size .3))
        (node (data (color shade) (size .3)))
        (node (data (color shade) (size .3)))
        (node (data (color shade) (size .3)))))

```




Figure 5: A generative model represented as a probabilistic program along with samples from the program.

branch function recursively connects a series of small nodes together and ends in a call to the flower function, passing it one of two colors. The flower function creates a node with three “petal” nodes each of the same color. Looking at this program we see how patterns such as a flower having three petals and the body having a large base are reflected in the program structure. The compositionality of the language also captures relational patterns such as branches end in flowers. Now that we know many different patterns can be represented within a probabilistic program, the main question becomes how do we change the structure of our initial data incorporated program to reflect such patterns.

We formalize the notion of pattern as repeated computation. Under this definition pattern recognition becomes a search for sequences of operations that are the same or similar. For example in figure [?] the sequences:

```

(node (data (color 33) (size 0.3))
  (node (data (color 220) (size 0.3)))
  (node (data (color 224) (size 0.3)))
  (node (data (color 207) (size 0.3))))

```

and

```

(node (data (color 37) (size 0.3))
  (node (data (color 213) (size 0.3)))
  (node (data (color 207) (size 0.3)))
  (node (data (color 211) (size 0.3))))

```

are very similar and so they might be considered instantiations of the same pattern. We can transform a program to recognize such patterns by creating functions or lambda abstractions based on these common sequences of operations e.g. by creating a `flower` function:

```

(define (flower)
  (node (data (color 37) (size 0.3))
        (node (data (color 213) (size 0.3)))
        (node (data (color 207) (size 0.3)))
        (node (data (color 211) (size 0.3)))))

```

This might seem like a limited definition of pattern, but we will demonstrate its ability to capture interesting patterns. It is also worth contemplating the role of lambda abstraction in the lambda calculus and the expressiveness of this language despite (or perhaps because of) its simplicity.

4 Bayesian Search

The search over program space uses beam search as a guiding heuristic like in Bayesian model merging. There is much that can be done in the way of employing this heuristic in complex search strategies, but for simplicity we use a straight-forward beam search.

4.1 Data Abstractions

We use data abstractions for programs and functions created by merge operations on programs. A function created by a merge operation is called an abstraction (after lambda abstraction). It consists of a name, variables, and a pattern i.e. the s-expression that makes up the body of the function.

```
(define (make-abstraction pattern variables)
  (make-named-abstraction (sym FUNC-SYMBOL) pattern variables))
(define (make-named-abstraction name pattern variables)
  (list 'abstraction name variables pattern))

(define abstraction->name second)
(define abstraction->vars third)
(define abstraction->pattern fourth)
```

Here `sym` is used to create unique symbols in an orderly manner for use as function or variable names.

Programs represent the generative models we search over. They consist of a list of abstractions and a body.

```
(define (make-program abstractions body)
  (list 'program abstractions body))
(define program->abstractions second)
(define program->body third)
```

We use a wrapper around the programs to keep track of additional program information during search. The motivation for this additional information will become clearer after reading the next section on beam search, but the basic idea is if a program's semantics are preserved after a transformation, the likelihood does not need to be recomputed (an expensive computation). The additional information in the wrapper is bookkeeping used to exploit this fact during search.

```
(define (make-program+ program posterior log-likelihood log-prior semantics-preserved)
  (list 'program+ program posterior log-likelihood log-prior semantics-preserved))
(define program+>program second)
(define program+>posterior third)
(define program+>log-likelihood fourth)
(define program+>log-prior fifth)
(define program+>semantics-preserved sixth)
(define (program+>program-transform semantics-preserved program+ new-program)
  (make-program+ new-program
    (program+>posterior program+)
    (program+>log-likelihood program+)
    (program+>log-prior program+)
    semantics-preserved))
```

4.2 Beam Search

We use a basic beam search to explore the space of generative models. The posterior is used as the heuristic function in guiding search.

```

(define (beam-search data init-program beam-size depth)
  (let* ([top-transformations
          (sort-by-posterior
           data
           (beam-search-transformations data init-program beam-size depth))]
         [da (display-all "top transformations" top-transformations)])
    (if (null? top-transformations)
        init-program
        (program+>program (first top-transformations))))))

```

```

(define (beam-search-transformations data program beam-size depth)
  (let ([init-program+ (make-program+ program 0 0 0 #f)])
    (depth-iterated-transformations (lambda (programs+)
                                      (best-n data programs+ beam-size))
                                   init-program+ depth)))

```

```

(define (best-n data programs n)
  (max-take (sort-by-posterior data programs) n))

```

The main part of the search is performed by `depth-iterated-transformations`, which recursively applies program transformations to the best programs at a given search depth and then filters the results to get the best programs for the next depth.

```

(define (depth-iterated-transformations cfilter program+ depth)
  (let* ([da (display-all "CURRENT DEPTH: " depth)]
         [transformed-programs+ (apply-and-filter-transformations depth cfilter program+)]
         (delete '() (append transformed-programs+
                              (apply append (map (lambda (prog)
                                                    (depth-iterated-transformations cfilter prog (- depth 1)))
                                                  transformed-programs+))))))

```

We reduce the amount of computation required when choosing the best programs at each level of search by separating program transformations that preserve semantics from those that do not. Marking programs based on their transformation type allows us to reuse likelihood for a program that was created by a semantics preserving transformation.

```

(define (apply-and-filter-transformations depth cfilter program+)
  (if (= depth 0)
      '()
      (let* ([semantics-preserved-programs+ (apply-transformations
                                             program+
                                             semantic-preserving-transformations
                                             #t)]
             [semantics-changed-programs+ (apply-transformations
                                             program+
                                             semantic-changing-transformations
                                             #f)])
        (cfilter (append semantics-preserved-programs+ semantics-changed-programs+))))))

```

```

(define (apply-transformations program+ transformations semantics-preserving)
  (let* ([program (program+>program program+)]
         [transformed-programs (delete '()
                                         (concatenate
                                          (map (lambda (transform)
                                                (transform program #t))
                                               transformations)))]
         [transformed-programs+ (map (lambda (program)
                                       (transform program #t))
                                     transformed-programs)])

```

```

                                (program+>program-transform
                                semantics-preserving
                                program+
                                program))
                                transformed-programs))])
transformed-programs+))

```

The posterior distribution is estimated by combining an estimate of the likelihood with the computation of the prior.

```

(define (sort-by-posterior data programs+)
  (let* ([programs (map program+>program programs+)]
        [semantics-flags (map program+>semantics-preserved programs+)]
        [log-priors (map log-prior programs)]
        [da (display-all "log-priors: " log-priors "\n")]
        [log-likelihoods (map (lambda (prog+ semantics-flag)
                               (if semantics-flag
                                   (program+>log-likelihood prog+)
                                   (log-likelihood data (program+>program prog+) 10)))
                              programs+
                              semantics-flags)]
        [da (display-all "log-likelihoods: " log-likelihoods "\n")]
        [posteriors (map + log-priors log-likelihoods)]
        [new-programs+ (map make-program+
                             programs posteriors log-likelihoods log-priors semantics-flags)]
        [posteriors> (lambda (a b) (> (program+>posterior a) (program+>posterior b)))]])
  (list-sort posteriors> new-programs+)))

```

We use a prior based on program length like in Bayesian model merging.

$$P(M) \propto e^{-size(M)} \quad (1)$$

```

(define (log-prior program)
  (- (program-size program)))

```

This prior effectively biases the search towards smaller programs. A program’s size is the number of symbols in the function bodies as well as the main body.

```

(define (program-size program)
  (define (sexpr-size sexpr)
    (if (list? sexpr)
        (apply + (map sexpr-size sexpr))
        1))
  (define (abstraction-size abstraction)
    (sexpr-size (abstraction->pattern abstraction)))
  (let* ([abstraction-sizes (apply + (map abstraction-size (program->abstractions program)))]
        [body-size (sexpr-size (program->body program))])
    (+ abstraction-sizes body-size)))

```

The computation of the likelihood is the difficult part in estimating the posterior. Intuitively we can think of it as way to say how good a particular program is at producing a set of target data. This is important in search because it can give information on whether and/or how to adjust our hypothesis. In a deterministic setting where the programs do not have randomness we can often have an intuitive sense of whether a program is good or bad based on some metric we make up on the output and the data. Working in the probabilistic setting allows us to have a more standard way of deciding whether a program is good or bad. The main idea being a probabilistic program is a generative process and represents a distribution over possible data. Given the “right” random choices during the execution of the program the output will be the observed data. By knowing the probabilities of these choices, we can combine them based on the rules of probability to get a score for how likely our program is to generate

the observed data and this gives us a sense of whether the program is good or bad. There may be many possible settings for the random choices of a program and this can make determining which choices lead to the observed data difficult. There is an additional difficulty in that there could be many possible settings that lead to the observed data and we need to take all of them into account. We give an example of one possible way to approximate this computation for list-type data in the example below.

4.2.1 Estimating Likelihood for List Programs

The method used here is described for the sake of completeness, but it is not critical to the idea of Bayesian program merging and there are many ways computation of likelihood could be improved. In the case of programs that generate list structured data we can estimate the likelihood by splitting the data into the discrete topology of the list and its continuous items. This factors the problem in the following way:

$$P(ls|p) = \prod_{ls} P(l|p) \quad (2)$$

where ls stands for lists, our target data, the l stands for a single list, and p is the program.

```
(define (log-likelihood lists-data prog sample-size)
  (apply + (map (lambda (single-list)
                  (single-log-likelihood prog sample-size single-list))
                lists-data)))
```

The idea behind estimating the likelihood for a single tree, $P(l|p)$, is to evaluate the program several times forcing the computation to result in the target tree each time. Each evaluation gives a probability for a possible way of generating the tree from the model. This probability is the product of random choices made in the program during evaluation. Since there may be several possible ways to generate the tree from the program we average the different evaluations. In the code below `smc-core` (where `smc` stands for sequential Monte Carlo) is the part that forces evaluation of the program to the desired data. A detailed description of `smc-core` is beyond the scope of this report, but it can be thought of as an incremental forward sampling process. Due to the forward-sampling nature of this process, we separate random choices used in generating the continuous-valued parts of the tree (i.e. the color) from the random choices that influence the tree structure.

```
(define single-log-likelihood
  (lambda (program popsize tree)
    (let* ([new-program (replace-color program)]
           [model (eval (program->sexpr new-program))]
           [topology-scores+tree-parameters (compute-topology-scores+evaluate
                                              model tree popsize)]
           [topology-scores (first topology-scores+tree-parameters)]
           [trees-with-parameters (second topology-scores+tree-parameters)]
           [color-scores (map (lambda (tree-with-parameters)
                               (compute-color-score tree tree-with-parameters))
                             trees-with-parameters)]
           [scores (map + topology-scores color-scores)]
           [score (if (null? scores)
                      -inf.0
                      (apply log-sum-exp scores))])
      score)))
```

We take advantage of fact that color is drawn from a Gaussian to compute the probability of observed color values. We do this by modifying the color function to output the mean and variance for a particular node rather than a value drawn from a gaussian with these parameters.

```
(define (replace-color program)
  (define (color? sexpr)
    (tagged-list? sexpr 'color))
```

```

(define (return-parameters sexpr)
  '(list ,(second sexpr) 15))
(define (replace-in-abstraction abstraction)
  (make-named-abstraction (abstraction->name abstraction)
    (sexp-search color? return-parameters (abstraction->pattern abstraction))
    (abstraction->vars abstraction)))
(let* ([converted-abstractions (map replace-in-abstraction (program->abstractions program))]
      [converted-body (sexp-search color? return-parameters (program->body program))])
  (make-program converted-abstractions converted-body)))

```

The modified program is evaluated several times to generate trees whose topology (node structure) matches the observed tree and has the parameters for the Gaussian used to generate a color. During the evaluation of the modified program we also capture the probability for the choices made in generating the topology of the tree. The references to laziness are related to the implementation of the sequential Monte Carlo procedure and the details are not critical to the report.

```

(define (compute-topology-scores+evaluate model tree popsize)
  (let* ([lazified-tree (list->lazy-list tree)]
        [samples (smc-core
          (map list (iota (+ 1 (lazy-list-size lazified-tree)))) popsize 20
          (lambda (depth) (lambda () (let ((s (model)))
            (pair (lazy-topology-equal? s lazified-tree depth)
              (lambda () (first (lazy-list->list s depth)))))))]
        [samples (fold (lambda (s a)
          (if (member (mcmc-state->addrval s)
            (map (lambda (x) (mcmc-state->addrval x repeat-symbol)) a))
            a (pair s a))) '() samples])
        [topology-scores (map mcmc-state->score samples)]
        [generated-trees (map mcmc-state->query-value samples)])
    (list topology-scores generated-trees)))

```

We can use the Gaussian parameters for each node's color to determine the probability of the observed color values of a tree.

```

(define (compute-color-score tree tree-with-parameters)
  (if (null? tree)
    0
    (+ (single-color-score (node->color tree) (node->color tree-with-parameters))
      (apply + (map compute-color-score
        (node->children tree)
        (node->children tree-with-parameters))))))

(define (single-color-score x mean+variance)
  (log (normal-pdf (first x) (first mean+variance) (second mean+variance))))

```

5 Merge Operations

We described earlier how patterns in the program representation of data can be viewed as repeated computation. Here we describe how this idea manifests itself as specific program transformations used in exploring the search space.

5.1 Inverse Inlining

Inlining is the process of replacing function calls in a program with the body of the function called. Inverse inlining creates new functions based on patterns in subexpressions of a program and replaces these subexpressions with functions calls for the new function. We can think of this process as creating lambda abstractions that potentially compress the program by removing duplication in the code and

acts as a proxy for recognizing repeated computation. The following code fragment outlines this procedure where compressions finds the different (lambda) abstractions that can be formed by anti-unifying (partially matching) pairs of subexpressions in a condensed form of the program (only the bodies of the functions and the body of the program). Duplicate abstractions are filtered out then potential program transformations are created by inverse-inlining function calls for each of the abstractions.

```
(define (compressions program . nofilter)
  (let*
    ([condensed-program (condense-program program)]
     [abstractions (possible-abstractions condensed-program)]
     [compressed-programs (map (curry compress-program program) abstractions)]
     [program-size (size (program->sexpr program))]
     [valid-compressed-programs
      ...find the compressed-programs smaller than the original program
      if nofilter is present then all compressed-programs are returned
    ])
    valid-compressed-programs))

(define (condense-program program)
  '(@,(map abstraction->pattern (program->abstractions program))
    ,(program->body program)))

(define (possible-abstractions expr)
  (let*
    ([subexpr-pairs (list-unique-commutative-pairs (all-subexprs expr))]
     [abstractions (map-apply (curry anti-unify-abstraction expr) subexpr-pairs)])
    (filter-abstractions abstractions)))
```

Aside from the obvious correspondence between repetition in program syntax and repetition in computation, there are a few different reasons for believing this is a useful program transformation. In terms of Bayesian model merging we are merging the structure of the model, which potentially leads to models with better generalization properties [TODO cite stolcke et al]. Inverse-inlining can also be interpreted as doing something similar to finding partial symmetries in inverse-procedural modeling [add bokeloh citation], which was successful in capturing the patterns of various 3-d models.

5.1.1 Overview

The following example illustrates the inverse inlining transformation

Given the program in figure 5.1.1 a possible result of this transformation would be The first thing to

```
(define (inverse-inlining-example)
  (uniform-draw
    (node
      (node a (node a (node b) (node b)))
      (node a (node a (node c) (node c))))))
```



Figure 6: An example generative model before an inverse inlining transformation.

note is both programs have the same behavior, in that both return either (a (a (b) (b))) or (a (a (c) (c))) with equal probability.

The transformation can be described as refactoring subexpressions that partially match in a program with a function whose body is the common parts of the matching subexpressions. In the above example the subexpressions that partially match are (node a (node a (node b) (node b))) and (node a (node a (node c) (node c))). The common subexpression is (node a (node a (node x) (node

```
(begin
  (define (F1 V1 V2)
    (node a (node a (node V1) (node V2))))
  (uniform-draw (node (F1 b b) (F1 c c))))
```




Figure 7: An example generative model after an inverse inlining transformation along with samples and a decomposition of the model into its subfunctions and the arguments passed to the subfunction.




Figure 8: An expression represented as a tree.

y))), the function created using this common subexpression is F1 and the original subexpressions are refactored as (F1 b b) and (F1 c c).

An inverse inlining transformation can be created for each pair of subexpressions that have a partial match. In the case of (+ (+ 2 2) (- 2 5)) the following pairs of subexpressions have a partial match: [2,2], [(+ 2 2), (- 2 5)], [(+ 2 2), (+ (+ 2 2) (+ 2 5))], [(+ 2 5), (+ (+ 2 2) (+ 2 5))], the only subexpressions that do not have a partial match in this example.

5.1.2 Anti-unification

The process of finding a partial match between two expressions is called anti-unification. One way to understand the process is in terms of the syntax trees for the expressions. Every s-expression can be thought of as a tree where the lists and sublists of the s-expression make up the interior nodes and the primitive elements of the lists (e.g. symbols, numbers, etc.) are the leaves. The tree in figure 5.1.2 corresponds to the expression (+ (+ 2 2) (- 2 5)). Finding a partial match between two expressions can be thought of as finding a common subtree between their tree representations.

Anti-unification proceeds by recursively comparing two expressions, A and B (performed by **build-pattern**). If A and B are the same primitive that primitive is returned. If A and B are lists (a_1, \dots, a_n) and (b_1, \dots, b_n) of the same length then a list C (c_1, \dots, c_n) where each element c_i is the anti-unification a_i and b_i is returned. For all other cases when A and B do not match a variable is returned.

```
(define anti-unify
  (lambda (expr1 expr2)
    (begin
      (define variables '())
      (define (add-variable!)
        (set! variables (pair (sym (var-symbol)) variables)))
```

```

    (first variables))
  (let ([pattern (build-pattern expr1 expr2)])
    (list pattern (reverse variables))))))

(define (build-pattern expr1 expr2)
  (cond
    [(and (primitive? expr1) (primitive? expr2)) (if (equal? expr1 expr2) expr1 (add-variable!))]
    [(or (primitive? expr1) (primitive? expr2)) (add-variable!)]
    [(not (eqv? (length expr1) (length expr2))) (add-variable!)]
    [else
     (let ([unified-expr
            (map (lambda (subexpr1 subexpr2) (build-pattern subexpr1 subexpr2))
                 expr1 expr2)])
       unified-expr)]))

```

5.1.3 Anti-unification example

We illustrate the process of anti-unification on the expressions $(+ (+ 2 2) (- 2 5))$ and $(+ (- 2 3) 4)$. The first step is to compare the root of the trees and make sure we have lists of the same size, in this case they are both of size three so we have a matching roots and a partial match of $(* * *)$ where the $*$'s are yet to be determined. Now we recursively attempt to match the three subexpressions $+$ with $+$, $(+ 2 2)$ with $(- 2 3)$, and $(- 2 5)$ with 4 . Since $+$ and $+$ are both primitives that are the same they match, and our partial match is now $(+ * *)$. Comparing $(+ 2 2)$ and $(- 2 3)$ we see that they are both lists of size 3 and so they match giving us a total partial match of $(+ (* * *) *)$. Again we recursively match subexpressions of $(+ 2 2)$ and $(- 2 3)$ i.e. $+$ to $-$, 2 to 2 , and 2 to 3 . Since $+$ and $-$ are primitives that don't match we replace them with a variable to get a total partial match of $(+ (V1 * *) *)$.

Likewise after comparing 2 to 2 and 2 to 3 we'll get $(+ (V1 2 V2) *)$ as our partial match. In the final comparison of $(- 2 5)$ and 4 we check can see there is no match because one $(- 2 5)$ is a list and 4 is a primitive so the final result of our anti-unification is $(+ (V1 2 V2) V3)$.

5.1.4 Compressing programs

Now that we have created lambda abstractions from the partial matches between two subexpressions of some program, P , we can use these abstractions to compress P . The idea will be to take the pattern of the abstraction and replace any subexpressions of P that fit the pattern. A subexpression that fits the pattern is replaced by an application of the lambda abstraction. We first do this replacement for each of the functions in P and then the body, resulting in a program P' . After all the replacements are made, the lambda abstraction is inserted into the set of functions for P' and this is our refactored program.

```

(define (compress-program program abstraction)
  (let* ([compressed-abstractions
         (map (curry compress-abstraction abstraction)
              (program->abstractions program))]
         [compressed-body (replace-matches (program->body program) abstraction)])
    (make-program (pair abstraction compressed-abstractions) compressed-body)))

(define (compress-abstraction compressor compressesee)
  (make-named-abstraction (abstraction->name compressesee)
                          (replace-matches (abstraction->pattern compressesee) compressor)
                          (abstraction->vars compressesee)))

```

Finding and replacing pattern matches of an abstraction F in an expression E proceeds recursively by taking the body of F and trying to match it to the expression E . A match between the body of the abstraction and the expression is determined by unification, if a match exists then the application of F is returned where the arguments to F are refactored (if F has arguments). If the match does not hold

then E is returned with each of its subexpressions refactored. If E is a primitive and not a match it is returned.

```
(define (replace-matches s abstraction)
  (let ([unified-vars (unify s
                             (abstraction->pattern abstraction)
                             (abstraction->vars abstraction))])
    (if (false? unified-vars) ;;a function call for the abstraction is returned if there is a match
        (if (list? s)
            (map (lambda (si) (replace-matches si abstraction)) s)
            s)
        (pair (abstraction->name abstraction)
              (map (lambda (var) (replace-matches (rest (assq var unified-vars)) abstraction))
                   (abstraction->vars abstraction))))))
```

5.1.5 Compressing Program Example

In the case of `(+ (+ 2 2) (- 2 5))` the partial match we would get between the subexpression `[(+ (+ 2 2) (- 2 5)), (+ 2 2)]` is `(+ V1 V2)`. We refactor the original expression `(+ (+ 2 2) (- 2 5))` in terms of `(+ V1 V2)` by creating a function `(define (F1 V1 V2) (+ V1 V2))` and applying it in the expression. For example one application could be `(F1 (+ 2 2) (- 2 5))` and another could be `(+ (F1 2 2) (- 2 5))`. In general we'll apply the function everywhere possible and get a refactored program like so...

```
(+ (+ 2 2) (- 2 5))
=>
(begin
  (define (F1 V1 V2) (+ V1 V2))
  (F1 (F1 2 2) (- 2 5)))
```

The input to the refactoring procedure is a function, F, created from anti-unification and an expression, E, which will be refactored in terms of F. In the example above F was `(define (F1 V1 V2) (+ V1 V2))`, E would be `(+ (+ 2 2) (- 2 5))`, and the result of refactoring was

```
(begin
  (define (F1 V1 V2) (+ V1 V2))
  (F1 (F1 2 2) (- 2 5))).
```

In the example `(+ (+ 2 2) (- 2 5))` matches `(+ V1 V2)` so an application of F1 to refactored arguments `(+ 2 2)` and `(- 2 5)` is returned resulting in `(F1 (F1 2 2) (- 2 5))`.

5.1.6 Unification

Determining whether there is a match between a function body and an expression is known as unification. Earlier we described anti-unification which can be thought of as a process to create a pattern from two expressions, unification can be viewed as the opposite process of seeing if and how a given pattern fits onto an expression. The return value of unification is a list of assignments for the variables of the function that would make the function the same as the expression.

The input to unification is a function F, and an expression E. Unification occurs recursively by checking whether the body of F and E are lists of the same size. If they are the same size then unification returns the unification of each of the subexpressions. If they are not the same size or only one of them is a list unification returns false. In the case where both expressions being unified are primitives true is returned if they are equal and false otherwise. In the case where the function expression of the unification is a variable an assignment is returned i.e. the variable along with the other expression passed to unification. At the end a check is made to see if any unifications have returned false, in which case unification of F and E returns false. There is also a check that any variable that is repeated in F has the same value assigned to it for each place it appears. If this is not the case unification returns false. If unification is not false then an assignment for each unique variable of F is returned.

```

(define unify
  (lambda (s sv vars)
    (cond [(variable? sv) (if (eq? s 'lambda) #f (list (pair sv s)))]
          [(and (primitive? s) (primitive? sv)) (if (eqv? s sv) '() #f)]
          [(or (primitive? s) (primitive? sv)) #f]
          [(not (eqv? (length s) (length sv))) #f]
          [else
           (let ([assignments (map (lambda (si sj) (unify si sj vars)) s sv)])
             (if (any false? assignments)
                 #f
                 (check/remove-repeated (apply append assignments)))))]))

(define (variable? obj)
  (member obj vars))

(define (check/remove-repeated unified-vars)
  (let* ([repeated-vars (filter more-than-one (map
                                              (curry all-assoc unified-vars)
                                              (map first unified-vars)))]
         (if (and (all (map all-equal? repeated-vars)) (not (any false? unified-vars)))
             (delete-duplicates unified-vars)
             #f)))

```

5.1.7 Unification Example

The example listed earlier where the function `F` is `(define (F1 V1 V2) (+ V1 V2))` and the s-expression being refactored is `(+ (+ 2 2) (- 2 5))` is used to illustrate unification. Since `(+ (+ 2 2) (- 2 5))` and `(+ V1 V2)` are the same length unification is applied to the subexpression pairs `[+,+]`, `[(+ 2 2), V1]`, `[(- 2 5), V2]`.

Unification between `+` and `+` returns nothing since neither is a variable and they match. Unification between `(+ 2 2)` and `V1` returns the assignment of `(+ 2 2)` to `V1` and likewise for `(- 2 5)` and `V2`. So the function `F1` matches the s-expression `(+ (+ 2 2) (- 2 5))` with variable assignments `V1:=(+ 2 2)` and `V2:=(- 2 5)`.

If the expression being unified with `F1` had been `(- (+ 2 2) (- 2 5))` then unification between the outer `-` of the expression and the `+` of `F1` would have returned false and the unification would have failed.

5.1.8 Summary

Inverse-inlining is a useful program transformation that identifies repeated computation in a program by finding patterns in its syntax. These patterns in syntax correspond to patterns in the data. The process follows two main steps 1) create abstractions from common subexpressions in a program via anti-unification 2) compress the program with the abstractions by replacing instances of the abstraction with a function application via unification. Going back to our plant example we illustrate the process on the program in figure 3.0.2. A few of the abstractions found with anti-unification that create the smallest compressed programs are

```

((abstraction F1 (V1) (data (color V1) (size 0.3))))

((abstraction F1 (V1) (node (data (color V1) (size 0.3)))))

((abstraction F1 (V1 V2) (data (color V1) (size V2))))

((abstraction F1 (V1 V2 V3 V4 V5 V6)
  (node (data (color V1) (size V2))
    (node (data (color V3) (size 0.3))
      (node (data (color V4) (size 0.3)))
      (node (data (color V5) (size 0.3)))))

```

```

        (node (data (color V6) (size 0.3))))))
((abstraction F1 (V1 V2 V3 V4)
  (node (data (color V1) (size 0.3))
    (node (data (color V2) (size 0.3))
      (node (data (color V3) (size 0.3))
        (node (data (color V4) (size 0.3))))))

```

The following are the programs compressed using these abstractions.

```

(begin
  (define F1 (lambda (V1) (data (color V1) (size 0.3))))
  ((uniform-draw
    (list
      (lambda ()
        (node (data (color 70) (size 1))
          (node (F1 37)
            (node (F1 213)) (node (F1 207)) (node (F1 211))))))
      (lambda ()
        (node (data (color 43) (size 1))
          (node (data (color 47) (size 0.1))
            (node (F1 33)
              (node (F1 220)) (node (F1 224)) (node (F1 207))))))))))

```

```

(begin
  (define F1
    (lambda (V1) (node (data (color V1) (size 0.3)))))
  ((uniform-draw
    (list
      (lambda ()
        (node (data (color 70) (size 1))
          (node (data (color 37) (size 0.3)) (F1 213)
            (F1 207) (F1 211))))
      (lambda ()
        (node (data (color 43) (size 1))
          (node (data (color 47) (size 0.1))
            (node (data (color 33) (size 0.3)) (F1 220)
              (F1 224) (F1 207))))))))))

```

```

(begin
  (define F1 (lambda (V1 V2) (data (color V1) (size V2))))
  ((uniform-draw
    (list
      (lambda ()
        (node (F1 70 1)
          (node (F1 37 0.3) (node (F1 213 0.3))
            (node (F1 207 0.3)) (node (F1 211 0.3)))))
      (lambda ()
        (node (F1 43 1)
          (node (F1 47 0.1)
            (node (F1 33 0.3) (node (F1 220 0.3))
              (node (F1 224 0.3)) (node (F1 207 0.3))))))))))

```

```

(begin
  (define F1
    (lambda (V1 V2 V3 V4 V5 V6)
      (node (data (color V1) (size V2))
        (node (data (color V3) (size 0.3))

```



```

      (node (data (color V4) (size 0.3)))
      (node (data (color V5) (size 0.3)))
      (node (data (color V6) (size 0.3))))))
((uniform-draw
  (list (lambda () (F1 70 1 37 213 207 211))
        (lambda ()
          (node (data (color 43) (size 1))
                (F1 47 0.1 33 220 224 207)))))))
(begin
  (define F1
    (lambda (V1 V2 V3 V4)
      (node (data (color V1) (size 0.3))
            (node (data (color V2) (size 0.3)))
            (node (data (color V3) (size 0.3)))
            (node (data (color V4) (size 0.3))))))
    ((uniform-draw
      (list
        (lambda ()
          (node (data (color 70) (size 1))
                (F1 37 213 207 211)))
        (lambda ()
          (node (data (color 43) (size 1))
                (node (data (color 47) (size 0.1))
                      (F1 33 220 224 207))))))))))

```

The last abstraction captures the flower pattern we mentioned earlier. Intuitively it seems like we can capture even more structure and replace the variables for the petal colors with a fixed value since they are all similar. So rather than explaining the data as being drawn from several gaussians with slightly different means, we would like to say the data comes from only one gaussian with a single mean. We address this issue in the next section with another type of program transformation.

5.2 Deargument

Deargument is a program transformation that takes a function in a program and removes one of its arguments. The variable that is removed is redefined inside the function in terms of all the values it has been assigned (i.e. the values passed to the function for that variable during an application). We can create different program transformations by allowing multiple schemes for redefining a variable, via `replacement-function`, as a function of its instantiations.

```

(define (deargument replacement-function program abstraction variable)
  (let* ([new-abstraction (remove-abstraction-variable
                        replacement-function
                        program abstraction
                        variable)])
    (if (null? new-abstraction)
        '()
        (let* ([program-with-new-abstraction (program->replace-abstraction
                                          program new-abstraction)]
              [new-program (remove-application-argument
                            program-with-new-abstraction
                            abstraction
                            variable)])
          new-program))))

```

The abstraction whose variable is being removed keeps the same body, but the variable removed is now assigned a value within the body instead of having its value passed in as an argument.

```

(define (remove-abstraction-variable replacement-function program abstraction variable)

```

```

(let* ([variable-instances (find-variable-instances program abstraction variable)]
      [variable-definition (replacement-function program abstraction variable variable-instances)]
      (if (equal? variable-definition NO-REPLACEMENT)
          '()
          (let* ([new-pattern
                  '((lambda (,variable) ,(abstraction->pattern abstraction)) ,variable-definition)]
                  [new-variables (delete variable (abstraction->vars abstraction))])
            (make-named-abstraction (abstraction->name abstraction) new-pattern new-variables)))))

(define (program->abstraction-applications program target-abstraction)
  (define (target-abstraction-application? sexpr)
    (if (non-empty-list? sexpr)
        (if (equal? (first sexpr) (abstraction->name target-abstraction))
            #t
            #f)
        #f))
  (let* ([abstraction-patterns (map abstraction->pattern (program->abstractions program))]
        [possible-locations (pair (program->body program) abstraction-patterns)])
    (deep-find-all target-abstraction-application? possible-locations)))

(define (deep-find-all pred? sexpr)
  (filter pred? (all-subexprs sexpr)))

(define (all-subexprs t)
  (let loop ([t (list t)])
    (cond [(null? t) '()]
          [(primitive? (first t)) (loop (rest t))]
          [else (pair (first t) (loop (append (first t) (rest t))))])))

(define (find-variable-instances program abstraction variable)
  (let* ([abstraction-applications (program->abstraction-applications program abstraction)]
        [variable-position (abstraction->variable-position abstraction variable)]
        [variable-instances (map (curry ith-argument variable-position) abstraction-applications)])
    variable-instances))

(define (ith-argument i function-application)
  (list-ref function-application (+ i 1)))

After the abstraction has been adjusted, any place it was used (i.e. the function was applied) needs to
be changed so that no values are passed in to the variable that was removed.

(define (remove-application-argument program abstraction variable)
  (define (abstraction-application? sexpr)
    (if (non-empty-list? sexpr)
        (equal? (first sexpr) (abstraction->name abstraction))
        #f))
  (define (change-application variable-position application)
    (define (change-recursive-arguments argument)
      (if (abstraction-application? argument)
          (change-application variable-position argument)
          argument))
    (let* ([ith-removed (remove-ith-argument variable-position application)])
      (map change-recursive-arguments ith-removed)))

  (remove-ith-argument variable-position application))
(let* ([variable-position (abstraction->variable-position abstraction variable)]
      [program-sexpr (program->sexpr program)])
  [program-sexpr])

```

```

    [changed-sexpr
      (sexp-search abstraction-application?
                    (curry change-application variable-position)
                    program-sexpr)]
    [new-program (sexpr->program changed-sexpr)]]
  new-program))

```

```

(define (sexp-search pred? func sexp)
  (if (pred? sexp)
      (func sexp)
      (if (list? sexp)
          (map (curry sexp-search pred? func) sexp)
          sexp)))

```

This transformation is useful for generalizing a program to have recursive structure as well as dealing with continuous data types that may be distorted with noise, for example the color data of the plant examples.

5.2.1 Noisy data example

One issue with inverse-inlining is anything other than perfect equality results in the creation of a variable when matching two expressions. So if we have an expression such as `(+ 2 2)` and `(+ 2 2.01)` and we know there is some noise in the system then we may want to treat these expressions as if they were the same i.e. inverse inline them to `(define (F1) (+ 2 2))` or `(define (F1) (+ 2 2.01))` rather than `(define (F1 x) (+ 2 x))`. We can do this using a deargument transformation along with a replacement-function called `noisy-number-replacement` that replaces the variable with the mean of its instances.

```

(define (noisy-number-replacement program abstraction variable variable-instances)
  (if (all (map number? variable-instances))
      (let*
        ([instances-mean (mean variable-instances)])
        instances-mean)
      NO-REPLACEMENT))

```

One possible problem with this implementation is we may end up creating many program transformations that will end up having a low likelihood because the variable instances do not actually come from the same gaussian. Since the likelihood calculation is expensive we can try to avoid this by adding a test to see whether the numbers are close together. If the numbers are close then the program is transformed otherwise it is not. The down side is our definition of what is close is brittle and we may end up not transforming a program that should be transformed.

```

(define (noisy-number-replacement program abstraction variable variable-instances)
  (define (close? a b)
    (< (abs (- a b)) 20))
  (if (all (map number? variable-instances))
      (let* ([instances-mean (my-mean variable-instances)]
              [instance-close-to-mean (map (curry close? instances-mean) variable-instances)])
        (if (all instance-close-to-mean)
            instances-mean
            NO-REPLACEMENT))
      NO-REPLACEMENT))

```

Going back to the plant example, suppose we have the program in figure 5.2.1 Applying the deargument transformation on the `flower` function and variable `V2` would first result in identifying instances of `V2`, which are 213 and 220. The abstraction, `flower`, is now changed using `remove-abstraction-variable` to get:

```

(define flower

```

```

(begin
  (define flower
    (lambda (V1 V2 V3 V4)
      (node (data (color V1) (size 0.3))
        (node (data (color V2) (size 0.3)))
        (node (data (color V3) (size 0.3)))
        (node (data (color V4) (size 0.3))))))
  ((uniform-draw
    (list
      (lambda ()
        (flower 200 213 207 211))
      (lambda ()
        (flower 33 220 224 207)))))))

```




Figure 9: A program before a noisy-number-deargument transformation.

```

(lambda (V1 V3 V4)
  ((lambda (V2)
    (node (data (color V1) (size 0.3))
      (node (data (color V2) (size 0.3)))
      (node (data (color V3) (size 0.3)))
      (node (data (color V4) (size 0.3))))))
    216.5)))

```

The whole program is now adjusted to incorporate the new version of `flower` using `remove-application-argument`. This creates a potentially simpler model (if there were more applications of `flower` with a likelihood that is similar to the original, which can be observed in the data generated by the transformed model (see figure 5.2.1). Now if we had applied the deargument transformation on the `flower` function to variabe

```

(begin
  (define flower
    (lambda (V1 V3 V4)
      ((lambda (V2)
        (node (data (color V1) (size 0.3))
          (node (data (color V2) (size 0.3)))
          (node (data (color V3) (size 0.3)))
          (node (data (color V4) (size 0.3))))))
        216.5)))
  ((uniform-draw
    (list
      (lambda ()
        (flower 200 207 211))
      (lambda ()
        (flower 33 224 207)))))))

```




Figure 10: A program after a noisy-number-deargument transformation that generates data similar to the original.

V1 whose variable instances are not similar we would get the following abstraction. Which creates a program that generated data unlike the original program (see figure 5.2.1.

```
(define flower
  (lambda (V2 V3 V4)
    ((lambda (V1)
      (node (data (color V1) (size 0.3))
            (node (data (color V2) (size 0.3))
                  (node (data (color V3) (size 0.3))
                        (node (data (color V4) (size 0.3))))))
      116.5)))

(begin
  (define flower
    (lambda (V2 V3 V4)
      ((lambda (V1)
        (node (data (color V1) (size 0.3))
              (node (data (color V2) (size 0.3))
                    (node (data (color V3) (size 0.3))
                          (node (data (color V4) (size 0.3))))))
        116.5)))
    ((uniform-draw
      (list
        (lambda ()
          (flower 213 207 211))
        (lambda ()
          (flower 220 224 207)))))))
```



Figure 11: A program after a noisy-number-deargument transformation that generates data different from the original.

5.2.2 Same Variable example

An inverse-inline transformation creates an abstraction with variables where the variables represent parts of an expression that differ between specific abstraction instances. It might be the case that variables with different names in the abstraction really ought to be treated as the same variable i.e. we would like to allow a variable to occur in multiple places within an abstract expression. We can address this with a deargument transformation that uses the the following replacement function.

```
(define (same-variable-replacement program abstraction variable variable-instances)
  (let* ([possible-match-variables (delete variable (abstraction->vars abstraction))])
    (uniform-draw possible-match-variables)))
```

Like in the case of the noisy-number deargument transformation we may end up performing many transformations that perform poorly because the variables that are merged should not actually be treated as the same. We can attempt to filter these poor transformations by checking to see if the variables are similar before performing the transformation. The basic idea is to check whether two variables take on the same values in each application of the function and if they do, they can be considered as the same variable. Putting this in terms of the deargument transformation we start with some variable of an abstraction we are trying to remove and check the other variables to see whether they match. The first variable that matches is returned as the definition for the variable that is being removed.

```
(define (same-variable-replacement program abstraction variable variable-instances)
  (let* ([possible-match-variables (delete variable (abstraction->vars abstraction))])
    (find-matching-variable program abstraction variable-instances possible-match-variables)))
```

Variable matches are determined by seeing if the instances for the variables are the same. The downside of filtering transformations in this manner is the equality check of the variable instances may be too strict and so we might not merge variables that ought to be the same because they only differ by a small amount.

```
(define (find-matching-variable program abstraction variable-instances possible-match-variables)
  (if (null? possible-match-variables)
      NO-REPLACEMENT
      (let* ([hypothesis-variable (first possible-match-variables)]
              [hypothesis-instances (find-variable-instances program abstraction hypothesis-variable)])
        (if (equal? hypothesis-instances variable-instances)
            hypothesis-variable
            (find-matching-variable program abstraction variable-instances (rest possible-match-variables))))))
```

Going back to the plant example, suppose we have the abstraction from figure 5.2.1.

```
(define flower
  (lambda (V1 V2 V3 V4)
    (node (data (color V1) (size 0.3))
          (node (data (color V2) (size 0.3))
                (node (data (color V3) (size 0.3))
                      (node (data (color V4) (size 0.3)))))))
```

Applying the deargument transformation with the `same-variable-replacement` to variable V2 can give us a program that generates data similar to the original if the V2 is matched to a variable that is a “petal” e.g. V3 (see figure 5.2.2). If the variable does not match a petal e.g. V1 then we get a program

```
(begin
  (define flower
    (lambda (V1 V3 V4)
      ((lambda (V2)
         (node (data (color V1) (size 0.3))
               (node (data (color V2) (size 0.3))
                     (node (data (color V3) (size 0.3))
                           (node (data (color V4) (size 0.3))))))
         V3)))
    (uniform-draw
     (list
      (lambda ()
        (flower 200 207 211))
      (lambda ()
        (flower 33 224 207)))))))
```



Figure 12: A program after a same-variable-deargument transformation that generates data similar to the original.

that generates data unlike the original program (see figure 5.2.2).

5.2.3 Recursion example

We illustrate how recursive patterns can be discovered using deargument and a `replacement-function` called `uniform-replacement` that redefines a variable as a uniform-draw over its instantiations.

```

(begin
  (define flower
    (lambda (V2 V3 V4)
      ((lambda (V1)
         (node (data (color V1) (size 0.3))
              (node (data (color V2) (size 0.3)))
              (node (data (color V3) (size 0.3)))
              (node (data (color V4) (size 0.3)))))
        V2)))
  ((uniform-draw
    (list
      (lambda ()
        (flower 213 207 211))
      (lambda ()
        (flower 220 224 207))))))

```



Figure 13: A program after a same-variable-deargument transformation that generates data unlike the original.

```

(define (uniform-replacement program abstraction variable variable-instances)
  '((uniform-draw (list ,@(map thunkify variable-instances)))))

```

We delay the evaluation of the `variable-instances` expressions by turning each into a thunk and performing an application on the thunk returned by the `uniform-draw`. Again, we can make a trade-off between efficiency (reducing the number of transformations examined) and search completeness by filtering out transformations that will probably result in undesirable programs. In the case of recursion we can ignore programs that will not produce a recursive pattern i.e. none of the variable instances is an application of the abstraction.

```

(define (uniform-replacement program abstraction variable variable-instances)
  (if (any (abstraction-application? abstraction sexpr))
      '((uniform-draw (list ,@(map thunkify variable-instances)))))
      NO-REPLACEMENT))

(define (abstraction-application? abstraction sexpr)
  (if (non-empty-list? sexpr)
      (if (equal? (first sexpr) (abstraction->name abstraction))
          #t
          #f)
      #f))

```

The program we transform is `(begin (node (node a)))` and we can apply the inverse inlining transformation to get the following: Now we'll apply deargument and remove `F1`'s argument. The first

```

(begin
  (define (F1 x)
    (node x))
  (F1 (F1 a)))

```



Figure 14: A program before a deargument transformation.

step is to change the definition of F1 (via `remove-abstraction-variable`) so that x is drawn from a distribution of past instances of the argument like so:

```
(begin
  (define (F1 x)
    (let ([x (uniform-draw (list (lambda () a) (lambda () (F1 a))))])
      (node x))
    (F1 (F1 a))))
```

Here the instances of x (i.e. what was passed into the function F1) are '(F1 a). The final step is to remove the argument from F1 and any applications of F1 resulting in the program using `remove-application-argument`.

```
(begin
  (define (F1)
    (let ([x ((uniform-draw (list (lambda () a) (lambda () (F1))))])
      (node x))
    (F1))
```



Figure 15: A program after a deargument transformation.

5.3 Deargument and Program Induction

In some sense the problem we are trying to address with the deargument transformation is the program induction problem itself. We can view the values for each variable as data generated by some process we would like to identify i.e. there is some sort of common computation between variable values that we would like to represent as a program. In the case of the noisy-number transformation we can view the generation of values for a single variable as coming from the same process, here we restrict the program representing this process to a Gaussian. Similary, in the case of the recursive transformation we can view the generation of values for a single variable as coming from a the same process, but we restrict the program representing this process to be the function being deargumented. In the case of the same variable transformation we can view the generation of values for multiple variables as coming from the same process and we implicitly use the program for one variable as the generating process for the other. It seems natural to ask whether we could reformulate the deargument transformation as recursive calls to the Bayesian program merging procedure in an attempt to find programs for generating the values of the variables. One can think of this as recursively squeezing out the randomness in the data where the abstractions created by inverse-inlining create a separation between the “structured” parts of the data (the fixed common sub-expressions) and the random parts of the data (the variables in the expression patterns). As an abstraction is applied one gets more data for the random parts and could conceivably attempt to learn a program to generate this data i.e. separate out even more structure.

6 Experiments

7 Discussion

We presented an approach to pattern recognition called Bayesian program merging. The main ideas were to rerepresent data as a program and then frame pattern recognition as finding repeated computation in the program. Finding repeated computation was performed through program transformations merged the structure of a program. The sequences of transformations made while searching were guided using the posterior of the program.

7.1 Noisy Data Constructors

We also allow for randomness in the data type constructors, which can lead to more compact representations of patterns in the presence of noise, which we will illustrate later.

It is worth noting the random nature of the color constructor. Randomness in data constructors can potentially allow for compact representations of patterns in the presence of noise. To get a sense of this idea look at figure ??.




Figure 16: Some plants.

x) (node x)) we might create the following generative model:

```
(multinomial-draw
 (node (data (color 20) (size .5)) (node (data (node 255) (size .5))))
 (node (data (node 20) (size .5)) (node (data (node 105) (size .5)))))
```

Which seems like a big penalty as far as model representation size goes since the probability of drawing the darker colored object may be incredibly small. By having a noisy constructor we can model the data more compactly as simply

```
(node (data (color 20) (size .5)) (node (data (node 255) (size .5))))
```

This example only hints at the benefits of using probabilistic data constructors and probabilistic programs in general with respect to representation, but understanding the full implications of such a design decision and its impact on program induction are left as future work.

7.2 Future Directions

Future directions include using search strategies more sophisticated than beam search and looking at better ways to compute the likelihood. Another open question is what happens when the data incorporation step is not so easy and training data is not directly computed in terms of some algebraic data type and whether one can impose some semblance of structure on unstructured data in order to use these methods for learning. There are many other barriers to overcome before probabilistic program induction can compete with state-of-the-art machine learning algorithms on real world problems, but the increased potential for capturing rich patterns and less dependence on human engineering make research in the subject a worthy pursuit.

References